Example 6b: Modeling Change over Time Using Piecewise Trends

(complete data, syntax, and output available for SAS, STATA, and R electronically)

These data for these example models come from Hoffman (2015) chapter 6. We will be examining change in response time (RT) in milliseconds over six practice sessions (balanced time) to a measure of processing speed in a sample of 101 older adults. This Example 6b builds on the polynomial time slopes (previously examined in Example 6a) to examine piecewise slopes for initial change and later change. Stay tuned for "truly" nonlinear exponential models (Example 6c using SAS only) and the use of log-transformed time to approximate a truly nonlinear exponential model (Example 6d).

SAS Syntax for Data Import and Manipulation:

```
* Define global variable for file location to be replaced in code below;
%LET filesave=C:\Dropbox\21_PSQF7375_Longitudinal\PSQF7375_Longitudinal_Example6;
* Location for SAS files for these models (uses macro variable filesave);
LIBNAME filesave "&filesave.";

DATA work.Example6; SET filesave.SAS_Chapter6;
* Create predictors for piecewise time models assuming BALANCED DATA;
* slope12 and slope26 for 2 Direct Slopes model, time1 for Slope + Deviation Slope model;
* Both models have intercept at session 1, breakpoint at session 2;
       IF session=1 THEN DO; slope12=0; slope26=0; END;
  ELSE IF session=2 THEN DO; slope12=1; slope26=0; END;
  ELSE IF session=3 THEN DO; slope12=1; slope26=1; END;
  ELSE IF session=4 THEN DO; slope12=1; slope26=2; END;
  ELSE IF session=5 THEN DO; slope12=1; slope26=3; END;
  ELSE IF session=6 THEN DO; slope12=1; slope26=4; END;
  LABEL slope12="slope12: Early Practice Slope (Session 1-2)"
        slope26="slope26: Later Practice Slope (Session 2-6)";
  time1=session-1; LABEL time1="time1: Session (0=1)" ;
* Create session dummy codes for testing means model absolute fit in REML;
  IF session=1 THEN s1=1; ELSE s1=0;
  IF session=2 THEN s2=1; ELSE s2=0;
  IF session=3 THEN s3=1; ELSE s3=0;
  IF session=4 THEN s4=1; ELSE s4=0;
  IF session=5 THEN s5=1; ELSE s5=0;
  IF session=6 THEN s6=1; ELSE s6=0;
RUN;
```

STATA Syntax for Data Import and Manipulation:

```
// Define global variable for file location to be replaced in code below
   global filesave "C:\Dropbox\21_PSQF7375_Longitudinal\PSQF7375_Longitudinal_Example6"
// Import chapter 6 stacked data and create time predictor variables
   use "$filesave\STATA_Chapter6.dta", clear

// Create predictors for piecewise time models assuming BALANCED DATA
// slope12 and slope26 for 2 Direct Slopes model, time1 for Slope + Deviation Slope model
// Both models have intercept at session 1, breakpoint at session 2
   gen slope12 = session
   recode slope12 (1=0) if session==1
   recode slope12 (2=1) if session==2
   recode slope12 (3=1) if session==3
   recode slope12 (4=1) if session==4
   recode slope12 (5=1) if session==5
   recode slope12 (6=1) if session==6
   gen slope26 = session
   recode slope26 (1=0) if session==1
   recode slope26 (2=0) if session==2
   recode slope26 (3=1) if session==3
   recode slope26 (4=2) if session==4
   recode slope26 (5=3) if session==5
   recode slope26 (6=4) if session==6
   label variable slope12 "slope12: Early Practice Slope (Session 1-2)"
   label variable slope26 "slope26: Later Practice Slope (Session 2-6)"
   gen time1=session-1
   label variable time1 "time1: Session (0=1)"
```

```
// Create session dummy codes for testing means model absolute fit in REML
   gen s4=0
   gen s5=0
   gen s6=0
   replace s4=1 if session==4
   replace s5=1 if session==5
   replace s6=1 if session==6
```

R Syntax for Data Import and Manipulation:

```
# Define variables for working directory and data name
filesave = "C:\\Dropbox/21_PSQF7375_Longitudinal/PSQF7375_Longitudinal_Example6/"
filename = "SAS_Chapter6.sas7bdat"
setwd(dir=filesave)

# Import chapter 6 stacked data with labels
Example6 = read_sas(data_file=paste0(filesave,filename))
# Convert to data frame as data frame without labels to use for analysis
Example6 = as.data.frame(Example6)
# Sort data by PersonID (needed for correct RCOV matrix)
Example6 = sort_asc(Example6,PersonID,wave)

# Create predictors for piecewise time models assuming BALANCED DATA
# slope12 and slope26 for 2 Direct Slopes model, time1 for Slope + Deviation Slope model
# Both models have intercept at session 1, breakpoint at session 2
Example6$slope12=Example6$session
Example6$slope12[which(Example6$session==1)]=0
Example6$slope12[which(Example6$session==2)]=1
Example6$slope12[which(Example6$session==3)]=1
Example6$slope12[which(Example6$session==4)]=1
Example6$slope12[which(Example6$session==5)]=1
Example6$slope12[which(Example6$session==6)]=1
Example6$slope26=Example6$session
Example6$slope26[which(Example6$session==1)]=0
Example6$slope26[which(Example6$session==2)]=0
Example6$slope26[which(Example6$session==3)]=1
Example6$slope26[which(Example6$session==4)]=2
Example6$slope26[which(Example6$session==5)]=3
Example6$slope26[which(Example6$session==6)]=4


# Center time predictor for polynomial time models
Example6$time1=Example6$session-1

# Create session dummy codes for testing means model absolute fit in REML
Example6$s4=0
Example6$s5=0
Example6$s6=0
Example6$s4[which(Example6$session==4)]=1
Example6$s5[which(Example6$session==5)]=1
Example6$s6[which(Example6$session==6)]=1
```

Model 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Slope12}_{ti}\right) + \beta_{2i}\left(\text{Slope26}_{ti}\right) + e_{ti}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + U_{0i}$

Slope12: $\beta_{1i} = \gamma_{10}$

Slope26: $\beta_{2i} = \gamma_{20}$

Session	1	2	3	4	5	6
Early Practice → Slope12 =	0	1	1	1	1	1
Later Practice → Slope26 =	0	0	1	2	3	4

```
TITLE1 "SAS Ch 6: 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = slope12 slope26 / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=PersonID;
     REPEATED session / R TYPE=VC SUBJECT=PersonID;
     ODS OUTPUT CovParms=CovFix12Fix26 InfoCrit=FitFix12Fix26; * Save for pseudo-R2 and LRT;
```

```
     * Get conditional mean per occasion from values of time predictors, also slope difference;
     ESTIMATE "Intercept at Session=1 Time=0"           intercept 1 slope12 0 slope26 0;
     ESTIMATE "Intercept at Session=2 Time=1"           intercept 1 slope12 1 slope26 0;
     ESTIMATE "Intercept at Session=3 Time=2"           intercept 1 slope12 1 slope26 1;
     ESTIMATE "Intercept at Session=4 Time=3"           intercept 1 slope12 1 slope26 2;
     ESTIMATE "Intercept at Session=5 Time=4"           intercept 1 slope12 1 slope26 3;
     ESTIMATE "Intercept at Session=6 Time=5"           intercept 1 slope12 1 slope26 4;
     ESTIMATE "Difference of fixed slope12 vs slope26"  slope12 -1 slope26 1;
RUN;
TITLE1 "Calculate pseudo R2 -- variance accounted for by fixed slope12 and fixed slope26";
%PseudoR2(NCov=2, CovFewer=CovEmpty, CovMore=CovFix12Fix26); TITLE1;
```

```
display "STATA Ch 6: 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model"
mixed rt c.slope12 c.slope26, || personid: , variance reml covariance(unstructured) ///
          residuals(independent,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      estat ic, n(101)                            // AIC and BIC
      estat wcorrelation, covariance              // V matrix
      estat wcorrelation                          // VCORR matrix
      // Get conditional mean per occasion from values of time predictors, also slope difference
      lincom _cons*1 + c.slope12*0 + c.slope26*0  // Intercept at Session=1
      lincom _cons*1 + c.slope12*1 + c.slope26*0  // Intercept at Session=2
      lincom _cons*1 + c.slope12*1 + c.slope26*1  // Intercept at Session=3
      lincom _cons*1 + c.slope12*1 + c.slope26*2  // Intercept at Session=4
      lincom _cons*1 + c.slope12*1 + c.slope26*3  // Intercept at Session=5
      lincom _cons*1 + c.slope12*1 + c.slope26*4  // Intercept at Session=6
      lincom c.slope12*-1 + c.slope26*1, small    // Difference of fixed slope12 vs slope26
      estimates store FitFix12Fix26               // Save for LRT
```

```
print("R Ch 6 4a: Fixed Slope12, Fixed Slope26, Random Intercept Model")
Fix12Fix26 = lmer(data=Example6, REML=TRUE, formula=rt~1+slope12+slope26+(1|PersonID))
print("Show results using Satterthwaite DDF including -2LL as deviance")
summary(Fix12Fix26, ddf="Satterthwaite"); llikAIC(Fix12Fix26, chkREML=FALSE)
print("Get conditional mean per occasion from values of time predictor, also slope difference")
print("Intercept at Session=1 Time=0"); contest1D(Fix12Fix26, ddf="Satterthwaite", L=c(1,0,0))
print("Intercept at Session=2 Time=1"); contest1D(Fix12Fix26, ddf="Satterthwaite", L=c(1,1,0))
print("Intercept at Session=3 Time=2"); contest1D(Fix12Fix26, ddf="Satterthwaite", L=c(1,1,1))
print("Intercept at Session=4 Time=3"); contest1D(Fix12Fix26, ddf="Satterthwaite", L=c(1,1,2))
print("Intercept at Session=5 Time=4"); contest1D(Fix12Fix26, ddf="Satterthwaite", L=c(1,1,3))
print("Intercept at Session=6 Time=5"); contest1D(Fix12Fix26, ddf="Satterthwaite", L=c(1,1,4))
print("Difference of fixed slope12 vs slope26"); contest1D(Fix12Fix26, ddf="Satterthwaite", L=c(0,-1,1))
```

SAS Output:

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	34098					
2		34098				
3			34098			
4				34098		
5					34098	
6						34098

This **level-1 R matrix VC structure** (equal variance over time, no covariance of any kind) will be used repeatedly as we add random piecewise time slopes to the model.

Estimated G Matrix

Row	Effect	PersonID	Col1
1	Intercept	101	202683

This **level-2 G matrix** (always **unstructured**) currently holding only the random intercept variance will be expanded as we add random effects.

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**236781**	202683	202683	202683	202683	202683
2	202683	**236781**	202683	202683	202683	202683
3	202683	202683	**236781**	202683	202683	202683
4	202683	202683	202683	**236781**	202683	202683
5	202683	202683	202683	202683	**236781**	202683
6	202683	202683	202683	202683	202683	**236781**

The marginal **V** matrix still has a compound symmetry pattern because we have not yet added to the model for the variance (still only a random intercept variance in **G**).

Estimated V Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8560	0.8560	0.8560	0.8560	0.8560
2	0.8560	1.0000	0.8560	0.8560	0.8560	0.8560
3	0.8560	0.8560	1.0000	0.8560	0.8560	0.8560
4	0.8560	0.8560	0.8560	1.0000	0.8560	0.8560
5	0.8560	0.8560	0.8560	0.8560	1.0000	0.8560
6	0.8560	0.8560	0.8560	0.8560	0.8560	1.0000

VCORR now provides a "conditional" ICC after controlling for fixed linear time. If the empty model ever returns ICC=0, report this conditional ICC instead.

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	PersonID	202683	29470	6.88	<.0001	Level-2 random intercept variance of U_{0i}
Session	PersonID	34098	2150.11	15.86	<.0001	Level-1 residual variance of e_{ti}

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	805.80	<.0001

This LRT is for the conditional ICC > 0.

Information Criteria

Neg2LogLike	**Parms**	AIC	AICC	HQIC	BIC	CAIC
8382.7	**2**	8386.7	8386.7	8388.8	8391.9	8393.9

Only variance model parameters are "counted" as parms using REML.

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	1961.89	48.4187	129	40.52	<.0001	gamma00
slope12	-163.64	23.2415	503	-7.04	<.0001	gamma10
slope26	-32.8932	5.8104	503	-5.66	<.0001	gamma20

Are the fixed piecewise slopes significant? How do we know?

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	**g = gamma fixed effect**
Intercept at Session=1 Time=0	1961.89	48.4187	129	40.52	<.0001	g00(1)+ g10(0)+ g20(0)
Intercept at Session=2 Time=1	1798.25	47.0035	115	38.26	<.0001	g00(1)+ g10(1)+ g20(0)
Intercept at Session=3 Time=2	1765.36	45.9134	104	38.45	<.0001	g00(1)+ g10(1)+ g20(1)
Intercept at Session=4 Time=3	1732.46	45.5443	101	38.04	<.0001	g00(1)+ g10(1)+ g20(2)
Intercept at Session=5 Time=4	1699.57	45.9134	104	37.02	<.0001	g00(1)+ g10(1)+ g20(3)
Intercept at Session=6 Time=5	1666.68	47.0035	115	35.46	<.0001	g00(1)+ g10(1)+ g20(4)
Difference of slope12 vs slope26	130.75	26.6265	503	4.91	<.0001	g20 – g10

Session	**1**	**2**	**3**	**4**	**5**	**6**
Early Practice → Slope12 =	**0**	**1**	**1**	**1**	**1**	**1**
Later Practice → Slope26 =	**0**	**0**	**1**	**2**	**3**	**4**



```
Calculate pseudo R2 -- variance accounted for by fixed slope12 and fixed slope26
PsuedoR2 (% Reduction) for CovEmpty vs. CovFix12Fix26
```

Name	CovParm	Subject	Estimate	StdErr	ZValue	ProbZ	PseudoR2
CovEmpty	UN(1,1)	PersonID	200883	29471	6.82	<.0001	.
CovEmpty	session	PersonID	44900	2825.63	15.89	<.0001	.
CovFix12Fix26	UN(1,1)	PersonID	202683	29470	6.88	<.0001	-0.00896
CovFix12Fix26	session	PersonID	34098	2150.11	15.86	<.0001	0.24058

Which variance did the fixed piecewise slopes explain? Why did the other variance increase?

Model 4b: Random Slope12, Fixed Slope26 Model

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Slope12}_{ti}\right) + \beta_{2i}\left(\text{Slope26}_{ti}\right) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Slope12: } \beta_{1i} = \gamma_{10} + U_{1i}$$

$$\text{Slope26: } \beta_{2i} = \gamma_{20}$$

```
TITLE1 "SAS Ch 6: 4b: Random Slope12, Fixed Slope26 Model";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = slope12 slope26 / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT slope12 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
     REPEATED session / R TYPE=VC SUBJECT=PersonID;
     ODS OUTPUT InfoCrit=FitRand12Fix26; * Save for LRT;
RUN;
TITLE1 "Calculate LRT -- does random slope12 improve model fit?";
%FitTest(FitFewer=FitFix12Fix26, FitMore=FitRand12Fix26); TITLE1;
```

```
display "STATA Ch 6: 4b: Random Slope12, Fixed Slope26 Model"
mixed rt c.slope12 c.slope26, || personid: slope12, variance reml covariance(unstructured) ///
           residuals(independent,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      estat ic, n(101)                                     // AIC and BIC
      estat recovariance, relevel(personid)                // G matrix
      estat recovariance, relevel(personid) correlation // GCORR matrix
      estat wcorrelation, covariance                       // V matrix
      estat wcorrelation                                   // VCORR matrix
      estimates store FitRand12Fix26         // Save for LRT
      lrtest FitRand12Fix26 FitFix12Fix26    // Does random slope12 improve model fit?
```

```
print("R Ch 6 4b: Random Slope12, Fixed Slope26 Model")
Rand12Fix26 = lmer(data=Example6, REML=TRUE, formula=rt~1+slope12+slope26+(1+slope12|PersonID))
print("Show results using Satterthwaite DDF including -2LL as deviance")
summary(Rand12Fix26, ddf="Satterthwaite"); llikAIC(Rand12Fix26, chkREML=FALSE)
print("Does random slope12improve model fit?")
ranova(Rand12Fix26, reduce.term=TRUE) # Remove random slope and covariance
```

SAS Output:

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	24168					
2		24168				
3			24168			
4				24168		
5					24168	
6						24168

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	277818	-69063
2	slope12	101	**-69063**	**59941**

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	1.0000	-0.5352
2	slope12	101	**-0.5352**	1.0000

This **level-2 G matrix** (always **unstructured**) still contains a random intercept variance (1,1), and we have added a random slope12 change variance (2,2), as well as a covariance between the random intercept and random slope12 for the same person (2,1). The intercept–slope12 correlation is shown in GCORR (2,1).

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**301985**	208755	208755	208755	208755	208755
2	208755	**223800**	199632	199632	199632	199632
3	208755	199632	**223800**	199632	199632	199632
4	208755	199632	199632	**223800**	199632	199632
5	208755	199632	199632	199632	**223800**	199632
6	208755	199632	199632	199632	199632	**223800**

How would we describe the marginal pattern of variance and covariances in the **V** matrix now?

Estimated V Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8030	0.8030	0.8030	0.8030	0.8030
2	0.8030	1.0000	0.8920	0.8920	0.8920	0.8920
3	0.8030	0.8920	1.0000	0.8920	0.8920	0.8920
4	0.8030	0.8920	0.8920	1.0000	0.8920	0.8920
5	0.8030	0.8920	0.8920	0.8920	1.0000	0.8920
6	0.8030	0.8920	0.8920	0.8920	0.8920	1.0000

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PersonID	277818	42741	6.50	<.0001	Level-2 random intercept variance of U_{0i}
UN(2,1)	PersonID	-69063	18932	-3.65	0.0003	Level-2 random intercept-slope12 covariance
UN(2,2)	PersonID	59941	12743	4.70	<.0001	Level-2 random slope12 change variance of U_{1i}
Session	PersonID	24168	1702.53	14.20	<.0001	Level-1 residual variance of e_{ti}

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
3	868.87	<.0001

This LRT tells us whether we need the full 2x2 **G** matrix (so it is not helpful).

Information Criteria

Neg2LogLike	**Parms**	AIC	AICC	HQIC	BIC	CAIC
8319.6	**4**	8327.6	8327.7	8331.8	8338.1	8342.1

Only variance model parameters are "counted" as parms using REML.

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	1961.89	54.6805	100	35.88	<.0001	gamma00
slope12	-163.64	31.2462	123	-5.24	<.0001	gamma10
slope26	-32.8932	4.8916	403	-6.72	<.0001	gamma20

Calculate LRT -- does random slope12 improve model fit?
Likelihood Ratio Test for FitFix12Fix26 vs. FitRand12Fix26

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitFix12Fix26	8382.7	2	8386.7	8391.9	.	.	.
FitRand12Fix26	8319.6	4	8327.6	8338.1	63.0739	2	2.0095E-14

Is the random slope12 significant? How do we know?

Model 4c: Random Slope12, Random Slope26 Model

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Slope12}_{ti}\right) + \beta_{2i}\left(\text{Slope26}_{ti}\right) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Slope12: } \beta_{1i} = \gamma_{10} + U_{1i}$$

$$\text{Slope26: } \beta_{2i} = \gamma_{20} + U_{2i}$$

```
TITLE1 "SAS Ch 6: 4c: Random Slope12, Random Slope26 Model";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = slope12 slope26 / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT slope12 slope26 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
     REPEATED session / R TYPE=VC SUBJECT=PersonID;
     ODS OUTPUT InfoCrit=FitRand12Rand26; * Save for LRT;
RUN;
TITLE1 "Calculate LRT -- does random slope26 improve model fit?";
%FitTest(FitFewer=FitRand12Fix26, FitMore=FitRand12Rand26); TITLE1;
```

```
display "STATA Ch 6: 4c: Random Slope12, Random Slope26 Model"
mixed rt c.slope12 c.slope26, || personid: slope12 slope26, variance reml covariance(unstructured) ///
           residuals(independent,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      estat ic, n(101)                                       // AIC and BIC
      estat recovariance, relevel(personid)                  // G matrix
      estat recovariance, relevel(personid) correlation // GCORR matrix
      estat wcorrelation, covariance                         // V matrix
      estat wcorrelation                                     // VCORR matrix
      estimates store FitRand12Rand26          // Save for LRT
      lrtest FitRand12Rand26 FitRand12Fix26    // Does random slope26 improve model fit?
```

```
print("R Ch 6 4c: Random Slope12, Random Slope26 Model -- reports convergence problem")
Rand12Rand26 = lmer(data=Example6, REML=TRUE, formula=rt~1+slope12+slope26+(1+slope12+slope26|PersonID))
print("Show results using Satterthwaite DDF including -2LL as deviance")
summary(Rand12Rand26, ddf="Satterthwaite"); llikAIC(Rand12Rand26, chkREML=FALSE)
print("Does random slope26 improve model fit?")
ranova(Rand12Rand26, reduce.term=TRUE) # Remove random slope and covariance
```

SAS Output:

Estimated R Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	17673					
2		17673				
3			17673			
4				17673		
5					17673	
6						17673

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	284312	-54270	-10644
2	slope12	101	-54270	63954	-1672.30
3	slope26	101	**-10644**	**-1672.30**	**2617.28**

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4025	-0.3902
2	slope12	101	-0.4025	1.0000	-0.1293
3	slope26	101	**-0.3902**	**-0.1293**	1.0000

The **level-2 G matrix** (always **unstructured**) still contains variances for the random intercept (1,1) and random slope12 change (2,2), as well as their covariance (2,1). We have now added a variance for the random slope26 change (3,3) and its covariances with the random intercept (3,1) and random slope12 (3,2) for the same person.

GCORR provides the corresponding correlations among the random effects.

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**301985**	230042	219399	208755	198111	187467
2	230042	**257400**	227410	215094	202778	190462
3	219399	227410	**235385**	208013	198314	188615
4	208755	215094	208013	**218604**	193850	186768
5	198111	202778	198314	193850	**207059**	184921
6	187467	190462	188615	186768	184921	**200747**

The marginal V matrix now predicts the variances to change in a **quadratic pattern (time^2) within each piece separately.**

Estimated V Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8251	0.8229	0.8125	0.7923	0.7614
2	0.8251	1.0000	0.9239	0.9068	0.8784	0.8379
3	0.8229	0.9239	1.0000	0.9170	0.8983	0.8677
4	0.8125	0.9068	0.9170	1.0000	0.9111	0.8916
5	0.7923	0.8784	0.8983	0.9111	1.0000	0.9070
6	0.7614	0.8379	0.8677	0.8916	0.9070	1.0000

The marginal **VCORR** matrix now predicts that the covariance changes over time in a time-dependent pattern **within each piece separately.**

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PersonID	284312	42731	6.65	<.0001	Level-2 random intercept variance of U_{0i}
UN(2,1)	PersonID	-54270	18230	-2.98	0.0029	Level-2 random intercept-slope12 covariance
UN(2,2)	PersonID	63954	13244	4.83	<.0001	Level-2 random slope12 change variance of U_{1i}
UN(3,1)	PersonID	-10644	3791.26	-2.81	0.0050	Level-2 random intercept-slope26 covariance
UN(3,2)	PersonID	-1672.30	2097.03	-0.80	0.4252	Level-2 random slope12-slope26 covariance
UN(3,3)	PersonID	2617.28	636.48	4.11	<.0001	Level-2 random slope26 change variance of U_{2i}
Session	PersonID	17673	1435.84	12.31	<.0001	Level-1 residual variance of e_{ti}

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
6	913.11	<.0001

This LRT tells us whether we need the full 3x3 **G** matrix (so it is not helpful).

Information Criteria

Neg2LogLike	**Parms**	AIC	AICC	HQIC	BIC	CAIC
8275.4	**7**	8289.4	8289.6	8296.8	8307.7	8314.7

Only variance model parameters are “counted” as parms using REML.

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	1961.89	54.6805	100	35.88	<.0001	gamma00
slope12	-163.64	30.2188	100	-5.42	<.0001	gamma10
slope26	-32.8932	6.5888	100	-4.99	<.0001	gamma20

Calculate LRT -- does random slope26 improve model fit?
Likelihood Ratio Test for FitRand12Fix26 vs. FitRand12Rand26

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRand12Fix26	8319.6	4	8327.6	8338.1	.	.	.
FitRand12Rand26	8275.4	7	8289.4	8307.7	44.2389	3	1.3427E-9

Is the random slope26 significant? How do we know?

<u>95% Random Effect Confidence Intervals that describe the *predicted* range of *individual* random effects:</u>

$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm \left(1.96*\sqrt{\text{Random Variance}}\right)$$

$$\text{Intercept 95\% CI} = \gamma_{00} \pm \left(1.96*\sqrt{\tau^2_{U_0}}\right) \rightarrow 1{,}961.9 \pm \left(1.96*\sqrt{284{,}312}\right) = 917 \text{ to } 3{,}007$$

$$\text{Slope12 95\% CI} = \gamma_{10} \pm \left(1.96*\sqrt{\tau^2_{U_1}}\right) \rightarrow -163.6 \pm \left(1.96*\sqrt{63{,}954}\right) = -659 \text{ to } 322$$

$$\text{Slope26 95\% CI} = \gamma_{20} \pm \left(1.96*\sqrt{\tau^2_{U_2}}\right) \rightarrow -32.9 \pm \left(1.96*\sqrt{2{,}617}\right) = -133 \text{ to } 67$$

So far we've examined one way to fit piecewise slopes models—two direct slopes that represent the change during each time period. Let's now examine an alternative specification—**slope + deviation slope**, which can be useful in examining individual differences in **differential change between time periods**.

Model 5a: Fixed Slope, Fixed Deviation Slope, Random Intercept Model (Equivalent to 4a)

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Time}_{ti}\right) + \beta_{2i}\left(\text{Slope26}_{ti}\right) + e_{ti}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + U_{0i}$

Time: $\beta_{1i} = \gamma_{10}$

Slope26: $\beta_{2i} = \gamma_{20}$

Session	1	2	3	4	5	6
Time → Slope16 =	0	1	2	3	4	5
Deviation → Slope26 =	0	0	1	2	3	4

```
TITLE1 "SAS Ch 6: 5a: Fixed Time, Fixed Slope26, Random Intercept Model";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = time1 slope26 / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT / G V VCORR TYPE=UN SUBJECT=PersonID;
     REPEATED session / R TYPE=VC SUBJECT=PersonID;
     ODS OUTPUT InfoCrit=FitFix16Fix26; * Save for LRT;
     * Get conditional mean per occasion from values of time predictors, also slope26;
     ESTIMATE "Intercept at Session=1 Time=0"            intercept 1 time1 0 slope26 0;
     ESTIMATE "Intercept at Session=2 Time=1"            intercept 1 time1 1 slope26 0;
     ESTIMATE "Intercept at Session=3 Time=2"            intercept 1 time1 2 slope26 1;
     ESTIMATE "Intercept at Session=4 Time=3"            intercept 1 time1 3 slope26 2;
     ESTIMATE "Intercept at Session=5 Time=4"            intercept 1 time1 4 slope26 3;
     ESTIMATE "Intercept at Session=6 Time=5"            intercept 1 time1 5 slope26 4;
     ESTIMATE "Fixed rate of change from session 2 to 6"   time1 1 slope26 1;
RUN; TITLE1;

display "STATA Ch 6: 5a: Fixed Time, Fixed Slope26, Random Intercept Model"
mixed rt c.time1 c.slope26, || personid: , variance reml covariance(unstructured) ///
           residuals(independent,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      estat ic, n(101)                            // AIC and BIC
      estat wcorrelation, covariance              // V matrix
      estat wcorrelation                          // VCORR matrix
      // Get conditional mean per occasion from values of time predictors, also slope26
      lincom _cons*1 + c.time1*0 + c.slope26*0      // Intercept at Session=1
      lincom _cons*1 + c.time1*1 + c.slope26*0      // Intercept at Session=2
      lincom _cons*1 + c.time1*2 + c.slope26*1      // Intercept at Session=3
      lincom _cons*1 + c.time1*3 + c.slope26*2      // Intercept at Session=4
      lincom _cons*1 + c.time1*4 + c.slope26*3      // Intercept at Session=5
      lincom _cons*1 + c.time1*5 + c.slope26*4      // Intercept at Session=6
      lincom c.time1*1 + c.slope26*1, small         // Fixed rate of change from session 2 to 6
      estimates store FitFix16Fix26   // Save for LRT

print("R Ch 6 5a: Fixed Time, Fixed Slope26, Random Intercept Model")
Fix16Fix26 = lmer(data=Example6, REML=TRUE, formula=rt~1+time1+slope26+(1|PersonID))
print("Show results using Satterthwaite DDF including -2LL as deviance")
summary(Fix16Fix26, ddf="Satterthwaite"); llikAIC(Fix12Fix26, chkREML=FALSE)
```

```
print("Get conditional mean per occasion from values of time predictors, also slope26")
print("Intercept at Session=1 Time=0"); contest1D(Fix16Fix26, ddf="Satterthwaite", L=c(1,0,0))
print("Intercept at Session=2 Time=1"); contest1D(Fix16Fix26, ddf="Satterthwaite", L=c(1,1,0))
print("Intercept at Session=3 Time=2"); contest1D(Fix16Fix26, ddf="Satterthwaite", L=c(1,2,1))
print("Intercept at Session=4 Time=3"); contest1D(Fix16Fix26, ddf="Satterthwaite", L=c(1,3,2))
print("Intercept at Session=5 Time=4"); contest1D(Fix16Fix26, ddf="Satterthwaite", L=c(1,4,3))
print("Intercept at Session=6 Time=5"); contest1D(Fix16Fix26, ddf="Satterthwaite", L=c(1,5,4))
print("Fixed rate of change from session 2 to 6"); contest1D(Fix16Fix26, ddf="Satterthwaite", L=c(0,1,1))
```

SAS Output (relevant tables only; is equivalent to model 4a):

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr > Z	
UN(1,1)	PersonID	202683	29470	6.88	<.0001	Level-2 random intercept variance of U_{0i}
session	PersonID	34098	2150.11	15.86	<.0001	Level-1 residual variance of e_{ti}

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
1	805.80	<.0001

This LRT is for the conditional ICC > 0 (in which ICC = .8560 for this model).

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8382.7	2	8386.7	8386.7	8388.8	8391.9	8393.9

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	1961.89	48.4187	129	40.52	<.0001	gamma00
time1	-163.64	23.2415	503	-7.04	<.0001	gamma10 (=slope16)
slope26	130.75	26.6265	503	4.91	<.0001	gamma20

Estimates

Label	Estimate	Standard Error	DF	t Value	Pr > |t|	**g = gamma fixed effect**
Intercept at Session=1 Time=0	1961.89	48.4187	129	40.52	<.0001	g00(1)+ g10(0)+ g20(0)
Intercept at Session=2 Time=1	1798.25	47.0035	115	38.26	<.0001	g00(1)+ g10(1)+ g20(0)
Intercept at Session=3 Time=2	1765.36	45.9134	104	38.45	<.0001	g00(1)+ g10(2)+ g20(1)
Intercept at Session=4 Time=3	1732.46	45.5443	101	38.04	<.0001	g00(1)+ g10(3)+ g20(2)
Intercept at Session=5 Time=4	1699.57	45.9134	104	37.02	<.0001	g00(1)+ g10(4)+ g20(3)
Intercept at Session=6 Time=5	1666.68	47.0035	115	35.46	<.0001	g00(1)+ g10(5)+ g20(4)
Rate of change from session 2 to 6	-32.8932	5.8104	503	-5.66	<.0001	g10 + g20



Model 5b: Random Slope, Fixed Deviation Slope Model

$$\text{Level 1: } y_{ti} = \beta_{0i} + \beta_{1i}\left(\text{Time}_{ti}\right) + \beta_{2i}\left(\text{Slope26}_{ti}\right) + e_{ti}$$

$$\text{Level 2: Intercept: } \beta_{0i} = \gamma_{00} + U_{0i}$$

$$\text{Time: } \beta_{1i} = \gamma_{10} + U_{1i}$$

$$\text{Slope26: } \beta_{2i} = \gamma_{20}$$

```
TITLE1 "SAS Ch 6: 5b: Random Time, Fixed Slope26 Model";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = time1 slope26 / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT time1 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
     REPEATED session / R TYPE=VC SUBJECT=PersonID;
     ODS OUTPUT InfoCrit=FitRand16Fix26; * Save for LRT;
RUN;
TITLE1 "Calculate LRT -- does random time1 slope improve model fit?";
%FitTest(FitFewer=FitFix16Fix26, FitMore=FitRand16Fix26); TITLE1;
```

```
display "STATA Ch 6: 5b: Random Time, Fixed Slope26 Model"
mixed rt c.time1 c.slope26, || personid: time1, variance reml covariance(unstructured) ///
           residuals(independent,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      estat ic, n(101)                                    // AIC and BIC
      estat recovariance, relevel(personid)               // G matrix
      estat recovariance, relevel(personid) correlation // GCORR matrix
      estat wcorrelation, covariance                      // V matrix
      estat wcorrelation                                  // VCORR matrix
      estimates store FitRand16Fix26         // Save for LRT
      lrtest FitRand16Fix26 FitFix16Fix26    // Does random time1 slope improve model fit?
```

```
print("R Ch 6 5b: Random Time, Fixed Slope26 Model")
Rand16Fix26 = lmer(data=Example6, REML=TRUE, formula=rt~1+time1+slope26+(1+time1|PersonID))
print("Show results using Satterthwaite DDF including -2LL as deviance")
summary(Rand16Fix26, ddf="Satterthwaite"); llikAIC(Rand16Fix26, chkREML=FALSE)
print("Does random time1 slope improve model fit?")
ranova(Rand16Fix26, reduce.term=TRUE) # Remove random slope and covariance
```

SAS Output:

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	254290	-12982
2	time1	101	**-12982**	**2346.46**

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2
1	Intercept	101	1.0000	-0.5315
2	time1	101	**-0.5315**	1.0000

This **level-2 G matrix** (always **unstructured**) still contains a random intercept variance (1,1), and we have added a random linear time slope variance (2,2), as well as a covariance between the random intercept and random linear time slope for the same person (2,1). The intercept–slope correlation is shown in GCORR (2,1).

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**280225**	241308	228325	215343	202361	189378
2	241308	**256606**	220036	209400	198764	188128
3	228325	220036	**237681**	203457	195168	186878
4	215343	209400	203457	**223449**	191571	185628
5	202361	198764	195168	191571	**213909**	184378
6	189378	188128	186878	185628	184378	**209063**

This random slope model predicts the same kind of **V** matrix as would a random linear time model—on the variance side, that's exactly what this is!

Estimated V Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8999	0.8847	0.8606	0.8265	0.7824
2	0.8999	1.0000	0.8910	0.8745	0.8484	0.8122
3	0.8847	0.8910	1.0000	0.8828	0.8656	0.8383
4	0.8606	0.8745	0.8828	1.0000	0.8762	0.8588
5	0.8265	0.8484	0.8656	0.8762	1.0000	0.8719
6	0.7824	0.8122	0.8383	0.8588	0.8719	1.0000

```
                 Covariance Parameter Estimates
                                    Standard         Z
Cov Parm     Subject     Estimate      Error     Value      Pr Z
UN(1,1)      PersonID      254290      37895      6.71    <.0001   Level-2 random intercept variance of U0i
UN(2,1)      PersonID      -12982    3620.52     -3.59    0.0003   Level-2 random intercept-time slope covariance
UN(2,2)      PersonID     2346.46     551.40      4.26    <.0001   Level-2 random linear time slope variance of U1i
Session      PersonID       25934    1827.00     14.20    <.0001   Level-1 residual variance of eti

  Null Model Likelihood Ratio Test
    DF    Chi-Square      Pr > ChiSq
     3        855.10          <.0001
```

This LRT tells us whether we need the full 2x2 **G** matrix (so it is not helpful).

```
                      Information Criteria
Neg2LogLike     Parms       AIC       AICC       HQIC        BIC       CAIC
     8333.4         4    8341.4     8341.4     8345.6     8351.8     8355.8

                    Solution for Fixed Effects
                          Standard
Effect        Estimate       Error      DF    t Value    Pr > |t|
Intercept      1961.89     52.6735     109      37.25      <.0001   gamma00
time1          -163.64     20.8345     467      -7.85      <.0001   gamma10
slope26         130.75     23.2213     403       5.63      <.0001   gamma20

Calculate LRT -- does random time1 slope improve model fit?
Likelihood Ratio Test for FitFix16Fix26 vs. FitRand16Fix26
                  Neg2Log
     Name          Like       Parms        AIC        BIC     DevDiff     DFdiff        Pvalue
FitFix16Fix26     8382.7        2       8386.7     8391.9       .           .               .
FitRand16Fix26    8333.4        4       8341.4     8351.8    49.3059        2        1.965E-11
```

This LRT tells us that the random linear time slope improves model fit.

Model 5c: Random Slope, Random Deviation Slope Model (Equivalent to 4c)

Level 1: $y_{ti} = \beta_{0i} + \beta_{1i}(\text{Time}_{ti}) + \beta_{2i}(\text{Slope26}_{ti}) + e_{ti}$

Level 2: Intercept: $\beta_{0i} = \gamma_{00} + U_{0i}$

Time: $\beta_{1i} = \gamma_{10} + U_{1i}$

Slope26: $\beta_{2i} = \gamma_{20} + U_{2i}$

```
TITLE1 "SAS Ch 6: 5c: Random Time, Random Slope26 Model";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = time1 slope26 / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT time1 slope26 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
     REPEATED session / R TYPE=VC SUBJECT=PersonID;
     ODS OUTPUT InfoCrit=FitRand16Rand26; * Save for LRT;
RUN;
TITLE1 "Calculate LRT -- does random slope26 improve model fit?";
%FitTest(FitFewer=FitRand16Fix26, FitMore=FitRand16Rand26); TITLE1;
```

```
display "STATA Ch 6: 5c: Random Time, Random Slope26 Model"
mixed rt c.time1 c.slope26, || personid: time1 slope26, variance reml covariance(unstructured) ///
           residuals(independent,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      estat ic, n(101)                                     // AIC and BIC
      estat recovariance, relevel(personid)                // G matrix
      estat recovariance, relevel(personid) correlation // GCORR matrix
      estat wcorrelation, covariance                       // V matrix
      estat wcorrelation                                   // VCORR matrix
      estimates store FitRand16Rand26           // Save for LRT
      lrtest FitRand16Rand26 FitRand16Fix26     // Does random slope26 improve model fit?
```

```
print("R Ch 6 5c: Random Time, Random Slope26 Model")
Rand16Rand26 = lmer(data=Example6, REML=TRUE, formula=rt~1+time1+slope26+(1+time1+slope26|PersonID))
```

```
print("Show results using Satterthwaite DDF including -2LL as deviance")
summary(Rand16Rand26, ddf="Satterthwaite"); llikAIC(Rand16Rand26, chkREML=FALSE)
print("Does random slope26 improve model fit?")
ranova(Rand16Rand26, reduce.term=TRUE) # Remove random slope and covariance
```

SAS Output (relevant tables only; is equivalent to model 4c):

The **level-2 G matrix** (always **unstructured**) still contains variances for the random intercept (1,1) and random linear time slope (2,2), **which is now just slope12 change (as in model 4c)**, as well as their covariance (2,1).

We have now added a variance for the **random slope26, which allows *differences* in change from sessions 2 to 6** (3,3), as well as its covariances with the random intercept (3,1) and random slope12 change (3,2) for the same person. **GCORR** provides the corresponding correlations among the random effects.

Estimated G Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	284312	-54270	43626
2	time1	101	-54270	63954	-65626
3	slope26	101	**43626**	**-65626**	**69916**

Estimated G Correlation Matrix

Row	Effect	PersonID	Col1	Col2	Col3
1	Intercept	101	1.0000	-0.4025	0.3094
2	time1	101	-0.4025	1.0000	-0.9814
3	slope26	101	**0.3094**	**-0.9814**	1.0000

Estimated V Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	**301985**	230042	219399	208755	198111	187467
2	230042	**257400**	227410	215094	202778	190462
3	219399	227410	**235385**	208013	198314	188615
4	208755	215094	208013	**218604**	193850	186768
5	198111	202778	198314	193850	**207059**	184921
6	187467	190462	188615	186768	184921	**200747**

The marginal V matrix now predicts the variances to change in a **quadratic pattern ($time^2$) within each piece separately, the same pattern as in model 4c.**

Estimated V Correlation Matrix for PersonID 101

Row	Col1	Col2	Col3	Col4	Col5	Col6
1	1.0000	0.8251	0.8229	0.8125	0.7923	0.7614
2	0.8251	1.0000	0.9239	0.9068	0.8784	0.8379
3	0.8229	0.9239	1.0000	0.9170	0.8983	0.8677
4	0.8125	0.9068	0.9170	1.0000	0.9111	0.8916
5	0.7923	0.8784	0.8983	0.9111	1.0000	0.9070
6	0.7614	0.8379	0.8677	0.8916	0.9070	1.0000

The marginal **VCORR** matrix now predicts that the covariance changes over time in a time-dependent pattern that differs **within each piece (epoch).**

Covariance Parameter Estimates

Cov Parm	Subject	Estimate	Standard Error	Z Value	Pr Z	
UN(1,1)	PersonID	284312	42731	6.65	<.0001	Level-2 random intercept variance of U_{0i}
UN(2,1)	PersonID	-54270	18230	-2.98	0.0029	Level-2 random intercept-slope12 covariance
UN(2,2)	PersonID	63954	13244	4.83	<.0001	Level-2 random time (=slope12 change) variance of U_{1i}
UN(3,1)	PersonID	43626	19049	2.29	0.0220	Level-2 random intercept-slope26 diff covariance
UN(3,2)	PersonID	-65626	14154	-4.64	<.0001	Level-2 random time-slope26 diff covariance
UN(3,3)	PersonID	69916	15434	4.53	<.0001	Level-2 random slope26 change diff variance of U_{2i}
Session	PersonID	17673	1435.84	12.31	<.0001	Level-1 residual variance of e_{ti}

Null Model Likelihood Ratio Test

DF	Chi-Square	Pr > ChiSq
6	913.11	<.0001

This LRT tells us whether we need the full 3x3 **G** matrix (so is not helpful).

Information Criteria

Neg2LogLike	Parms	AIC	AICC	HQIC	BIC	CAIC
8275.4	7	8289.4	8289.6	8296.8	8307.7	8314.7

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|	
Intercept	1961.89	54.6805	100	35.88	<.0001	gamma00
time1	-163.64	30.2188	100	-5.42	<.0001	gamma10
slope26	130.75	32.5530	100	4.02	0.0001	gamma20

```
Calculate LRT -- does random slope26 improve model fit?
Likelihood Ratio Test for FitRand16Fix26 vs. FitRand16Rand26
                      Neg2Log
     Name               Like      Parms        AIC        BIC    DevDiff    DFdiff       Pvalue
FitRand16Fix26        8333.4        4       8341.4     8351.8       .          .              .
FitRand16Rand26       8275.4        7       8289.4     8307.7    58.0069       3      1.5665E-12
```

This LRT tells us that the random time slope improves model fit.

95% Random Effect Confidence Intervals that describe the *predicted* range of *individual* random effects:

$$\text{Random Effect 95\% CI} = \text{fixed effect} \pm \left(1.96*\sqrt{\text{Random Variance}}\right)$$

$$\text{Intercept 95\% CI} = \gamma_{00} \pm \left(1.96*\sqrt{\tau^2_{U_0}}\right) \rightarrow 1{,}961.9 \pm \left(1.96*\sqrt{284{,}312}\right) = 917 \text{ to } 3{,}007$$

$$\text{Time 95\% CI} = \gamma_{10} \pm \left(1.96*\sqrt{\tau^2_{U_1}}\right) \rightarrow -163.6 \pm \left(1.96*\sqrt{63{,}954}\right) = -659 \text{ to } 322$$

$$\text{Slope26 95\% CI} = \gamma_{20} \pm \left(1.96*\sqrt{\tau^2_{U_2}}\right) \rightarrow 130.8 \pm \left(1.96*\sqrt{69{,}916}\right) = -338 \text{ to } 649$$

So how did we do? Let's compare model predictions in terms of means (top) and variances (bottom)? Note that models 4c and 5c are equivalent, so only 4c is shown.





Bonus Material: Testing Absolute Fit of Each Side of the Model When Using REML

As shown as Model 0, the saturated means, unstructured variance model is the best-fitting model for each side (means and variances). However, when using REML, we cannot do a model comparison against our random two-piece slopes model, because models cannot differ in their fixed effects for the −2LL (LRT) to be valid. Instead, we can test the absolute fit for each side of the model separately as shown next. (This will be included in the second edition of my textbook, in progress).

The absolute fit of the two-piece model for the means can be tested by mimicking a saturated means model using the *same random two-piece slopes* (i.e., holding the model for the variance constant). Because the change from sessions 1–2 is already predicted perfectly, the session dummy codes must be added to the second slope (i.e., that still has degrees of freedom for not fitting the change from 2–6 perfectly).

```
TITLE1 "SAS: Test Absolute Fit of Two-Piece Means Model (Using Random Two-Piece Variance Model)";
TITLE2 "Add 3 session dummy codes to SECOND PIECE to saturate the means model";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = time1 slope26 s4 s5 s6 / SOLUTION DDFM=Satterthwaite;
     RANDOM INTERCEPT time1 slope26 / G GCORR V VCORR TYPE=UN SUBJECT=PersonID;
     REPEATED session / R TYPE=VC SUBJECT=PersonID;
     CONTRAST "Does fixed two-piece slope reproduce saturated means?" s4 1, s5 1, s6 1 / CHISQ;
RUN; TITLE1; TITLE2;
```

```
display "STATA: Test Absolute Fit of Two-Piece Means Model (Using Random Two-Piece Variance Model)"
display "Add 3 session dummy codes to second piece to saturate the means model"
mixed rt c.time1 c.slope26 c.s4 c.s5 c.s6, ///
           || personid: time1 slope26, variance reml covariance(unstructured) ///
           residuals(independent,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      test (c.s4=0)(c.s5=0)(c.s6=0), small  // Does fixed two-piece slope reproduce saturated means?
```

```
print("R: Test Absolute Fit of Two-Piece Means Model (Using Random Two-Piece Variance Model)")
print("Add 3 session dummy codes to SECOND PIECE to saturate the means model")
PieceMean = lmer(data=Example6, REML=TRUE,
                 formula=rt~1+time1+slope26+s4+s5+s6+(1+time1+slope26|PersonID))
print("Show results using Satterthwaite DDF including -2LL as deviance")
summary(PieceMean, ddf="Satterthwaite"); llikAIC(PieceMean, chkREML=FALSE)
print("Does fixed two-piece slope reproduce saturated means?")
contestMD(PieceMean, ddf="Satterthwaite", L=rbind(c(0,0,0,1,0,0),c(0,0,0,0,1,0),c(0,0,0,0,0,1)))
```

SAS Output (relevant tables only):

Solution for Fixed Effects

Effect	Estimate	Standard Error	DF	t Value	Pr > |t|
Intercept	1961.89	54.6805	100	35.88	<.0001
time1	-146.72	31.3492	116	-4.68	<.0001
slope26	81.5832	41.6921	237	1.96	0.0515
s4	32.8995	32.3095	300	1.02	0.3094
s5	87.4166	49.3536	300	1.77	0.0775
s6	117.51	67.2576	300	1.75	0.0816

Because there are now **6 fixed effects for the 6 means**, this model is equivalent to **saturated means** (even if the two fixed piecewise slopes are largely uninterpretable). In an SEM context, this model would be specified by letting three of the observed occasions' intercepts be estimated (as discrepancies).

The multivariate Wald test using CONTRAST indicates that the 3 extra session contrasts did not improve model fit (which is good news here).

Contrasts

Label	Num DF	Den DF	Chi-Square	F Value	Pr > ChiSq	Pr > F
Does fixed two-piece slope reproduce saturated means?	3	300	4.74	1.58	0.1922	**0.1946**

The absolute fit of the random two-piece model for the variance can be tested against a UN variance model using the *same fixed two-piece slopes* (i.e., holding the model for the means constant):

```
TITLE1 "SAS: Test Absolute Fit of Two-Piece Variance Model (Using Fixed Two-Piece Means Model)";
TITLE2 "Change to Unstructured R matrix as variance model answer key";
PROC MIXED DATA=work.Example6 COVTEST NOCLPRINT NAMELEN=100 IC METHOD=REML;
     CLASS PersonID session;
     MODEL rt = time1 slope26 / SOLUTION DDFM=Satterthwaite;
     REPEATED session / R RCORR TYPE=UN SUBJECT=PersonID;
     ODS OUTPUT InfoCrit=FitFixPieceUN; * Save for LRT;
RUN; TITLE2;
TITLE1 "Calculate LRT -- does random two-piece slope reproduce UN variance model?";
%FitTest(FitFewer=FitRand16Rand26, FitMore=FitFixPieceUN); TITLE1;
```

```
display "STATA: Test Absolute Fit of Two-Piece Variance Model (Using Fixed Two-Piece Means Model)"
display "Change to Unstructured R matrix as variance model answer key"
mixed rt c.time1 c.slope26, || personid: , noconstant variance reml ///
          residuals(unstructured,t(session)) dfmethod(satterthwaite) dftable(pvalue)
      estimates store FitFixPieceUN         // Save for LRT
      lrtest FitFixPieceUN FitRand16Rand26  // Does random two-piece slope reproduce UN variance model?
```

```
print("R: Test Absolute Fit of Two-Piece Variance Model (Using Fixed Two-Piece Means Model)")
print("Change to Unstructured R matrix as variance model answer key in GLS")
FixPieceUN = gls(data=Example6, method="REML", model=rt~1+time1+slope26,
                 correlation=corSymm(form=~as.numeric(session)|PersonID),
                 weights=varIdent(form=~1|session))
print("Have to re-run random two-piece model using LME to get LRT")
RandPiecelme = lme(data=Example6, method="REML", rt~1+time1+slope26, random=~1+time1+slope26|PersonID)
anova(FixPieceUN,RandPiecelme) # anova does LRT using LME versions
```

SAS Output (relevant table only):

Calculate LRT -- does random two-piece slope reproduce UN variance model?
Likelihood Ratio Test for FitRand16Rand26 vs. FitRandUN

Name	Neg2Log Like	Parms	AIC	BIC	DevDiff	DFdiff	Pvalue
FitRand16Rand26	8275.4	7	8289.4	8307.7	.	.	.
FitRandUN	8259.6	21	8301.6	8356.5	15.7728	14	**0.32744**

What does this nonsignificant LRT result indicate about our random two-piece slopes model?